Class 4 : Programming in Mathematica Part #2

Note : Almost all of the examples below are taken from the Mathematica documentation center.

In this class we shall see how to handle, expressions, lists, functions, patterns (a bit). In the next class we shall see shall look at procedural programming in Mathematica.

Making Lists of Objects

```
{3, 5, 1}
```

{3, 5, 1}

```
{3, 5, 1}^2 + 1
```

{10, 26, 2}

```
{6, 7, 8} - {3.5, 4, 2.5}
```

{2.5, 3, 5.5}

```
Exp[%] // N
```

{12.1825, 20.0855, 244.692}

```
v = {2, 4, 3.1}
```

{2, 4, 3.1}

```
v / (v - 1)
```

$\left\{2, \frac{4}{3}, 1.47619\right\}$

Getting Pieces of Lists

```
t = {a, b, c, d, e, f, g}
```

{a, b, c, d, e, f, g}

```
Last[t]
```

g

```
t[[3]]
```

c

```
t[[3 ;; 6]]
```

{c, d, e, f}

```
t[[{1, 4}]]
```

```
{a, d}
```

```
Take[t, 3]
```

```
{a, b, c}
```

```
Take[t, -3]
```

```
{e, f, g}
```

```
Take[t, {3, 7, 2}]
```

```
{c, e, g}
```

```
Rest[t]
```

```
{b, c, d, e, f, g}
```

```
Drop[t, 3]
```

```
{d, e, f, g}
```

Defining Functions

```
f[x_] := x^2
```

```
f[a + 1]
```

$(1 + a)^2$

```
f[4]
```

```
16
```

```
? f
```

```
Global`f
```

$f[x\_] := x^2$

```
hump[x_, xmax_] := (x - xmax)^2 / xmax
```

```
2 + hump[x, 3.5]
```

$2 + 0.285714\ (-3.5 + x)^2$

```
hump[x_, xmax_] := (x - xmax)^4
```

```
? hump
```

```
Global`hump
```

$hump[x\_, xmax\_] := (x - xmax)^4$

```
Clear[hump]
```

Applying Functions to Parts of Expressions

```
Map[f, {a, b, c}]
```

```
{f[a], f[b], f[c]}
```

```
take2[list_] := Take[list, 2]
```

```
Map[take2, {{1, 3, 4}, {5, 6, 7}, {2, 1, 6, 6}}]
```

```
{{1, 3}, {5, 6}, {2, 1}}
```

```
Map[f, a + b + c]
```

```
f[a] + f[b] + f[c]
```

```
Map[Sqrt, g[x^2, x^3]]
```

$$g\left[\sqrt{x^2}\,,\ \sqrt{x^3}\,\right]$$

```
m = {{a, b}, {c, d}}
```

```
{{a, b}, {c, d}}
```

```
Map[f, m]
```

```
{f[{a, b}], f[{c, d}]}
```

```
Map[f, m, {2}]
```

```
{{f[a], f[b]}, {f[c], f[d]}}
```

```
Scan[Print, {a, b, c}]
```

```
a
b
c
```

```
Scan[Print, 1 + x^2, Infinity]
```

```
1
x
2
x^2
```

Applying Functions to Lists and Other Expressions

```
Apply[f, {a, b, c}]
```

```
f[a, b, c]
```

```
Apply[Times, {a, b, c}]
```

```
a b c
```

```
geom[list_] := Apply[Times, list]^(1 / Length[list])
```

```
Apply[List, a + b + c]
```

```
{a, b, c}
```

```
m = {{a, b, c}, {b, c, d}}
```

```
{{a, b, c}, {b, c, d}}
```

```
Apply[f, m]
```

```
f[{a, b, c}, {b, c, d}]
```

```
Apply[f, m, {1}]
```

```
{f[a, b, c], f[b, c, d]}
```

```
Apply[f, m, {0, 1}]
```

```
f[f[a, b, c], f[b, c, d]]
```

Adding, Removing and Modifying List Elements

```
Prepend[{a, b, c}, x]
```

```
{x, a, b, c}
```

```
Insert[{a, b, c}, x, 2]
```

```
{a, x, b, c}
```

```
Riffle[{a, b, c}, x]
```

```
{a, x, b, x, c}
```

```
?Riffle
```

Riffle[{e_1, e_2, …}, x] gives {e_1, x, e_2, x, …}.
Riffle[{e_1, e_2, …}, {x_1, x_2, …}] gives {e_1, x_1, e_2, x_2, …}.
Riffle[*list*, x, n] yields a list in which every n^{th} element is x.
Riffle[*list*, x, {i_{min}, i_{max}, n}] yields a list in which x appears if possible at positions i_{min}, $i_{min} + n$, $i_{min} + 2\,n$, … , i_{max}. ≫

```
ReplacePart[{a, b, c, d}, 3 -> x]
```

```
{a, b, x, d}
```

```
ReplacePart[{{a, b}, {c, d}}, {1, 2} -> x]
```

```
{{a, x}, {c, d}}
```

```
v = {a, b, c, d}
```

```
{a, b, c, d}
```

```
v[[3]] = x
```

```
x
```

```
v
```

```
{a, b, x, d}
```

```
m = {{a, b}, {c, d}}
```

```
{{a, b}, {c, d}}
```

```
m[[All, 1]] = {x, y}; m
```

```
{{x, b}, {y, d}}
```

```
m[[All, 1]] = 0; m
```

```
{{0, b}, {0, d}}
```

Manipulating Lists by Their Indices

```
{a, b, c, d}[[{1, 3}]]
```

```
{a, c}
```

```
(m = {{a, b, c}, {d, e, f}, {g, h, i}}) // TableForm
```

```
a b c
d e f
g h i
```

```
m[[{1, 3}, {1, 2}]] // TableForm
```

```
a b
g h
```

```
Extract[m, {{1, 3}, {1, 2}}]
```

```
{c, b}
```

```
m = {{a[1], a[2], b[1]}, {b[2], c[1]}, {{b[3]}}};
```

```
Position[m, b[_]]
```

```
{{1, 3}, {2, 1}, {3, 1, 1}}
```

```
Extract[m, %]
```

```
{b[1], b[2], b[3]}
```

```
Insert[{{a, b, c}, {d, e}}, x, {2, 1}]
```

```
{{a, b, c}, {x, d, e}}
```

```
Delete[%, {2, 1}]
```

```
{{a, b, c}, {d, e}}
```

```
IdentityMatrix[3]
```

```
{{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}
```

```
ReplacePart[%, {2, 2} -> x]
```

```
{{1, 0, 0}, {0, x, 0}, {0, 0, 1}}
```

Nested Lists

```
Table[x^i + j, {i, 2}, {j, 3}]
```

$$\left\{\{1+x,\ 2+x,\ 3+x\},\ \left\{1+x^2,\ 2+x^2,\ 3+x^2\right\}\right\}$$

```
Array[x^#1 + #2 &, {2, 3}]
```

$$\{\{1 + x, 2 + x, 3 + x\}, \{1 + x^2, 2 + x^2, 3 + x^2\}\}$$

```
Table[x^i + j, {i, 3}, {j, i}]
```

$$\{\{1 + x\}, \{1 + x^2, 2 + x^2\}, \{1 + x^3, 2 + x^3, 3 + x^3\}\}$$

```
Array[a, {2, 3}]
```

```
{{a[1, 1], a[1, 2], a[1, 3]}, {a[2, 1], a[2, 2], a[2, 3]}}
```

```
Flatten[%]
```

```
{a[1, 1], a[1, 2], a[1, 3], a[2, 1], a[2, 2], a[2, 3]}
```

```
ArrayFlatten[{{{{1}}, {{2, 3}}}, {{{4}, {7}}, {{5, 6}, {8, 9}}}}]
```

```
{{1, 2, 3}, {4, 5, 6}, {7, 8, 9}}
```

```
Array[a, {2, 2, 2}]
```

```
{{{a[1, 1, 1], a[1, 1, 2]}, {a[1, 2, 1], a[1, 2, 2]}},
 {{a[2, 1, 1], a[2, 1, 2]}, {a[2, 2, 1], a[2, 2, 2]}}}
```

```
Transpose[%, {3, 1, 2}]
```

```
{{{a[1, 1, 1], a[2, 1, 1]}, {a[1, 1, 2], a[2, 1, 2]}},
 {{a[1, 2, 1], a[2, 2, 1]}, {a[1, 2, 2], a[2, 2, 2]}}}
```

```
m = {{{a, b}, {c, d}}, {{e, f}, {g, h}, {i}}};
```

```
Map[f, m, {2}]
```

```
{{f[{a, b}], f[{c, d}]}, {f[{e, f}], f[{g, h}], f[{i}]}}
```

```
Apply[f, m, {2}]
```

```
{{f[a, b], f[c, d]}, {f[e, f], f[g, h], f[i]}}
```

Pure Functions

```
h[x_] := f[x] + g[x]
```

```
Map[h, {a, b, c}]
```

```
{f[a] + g[a], f[b] + g[b], f[c] + g[c]}
```

```
Map[f[#] + g[#] &, {a, b, c}]
```

```
{f[a] + g[a], f[b] + g[b], f[c] + g[c]}
```

```
Function[x, x^2]
```

$$\text{Function}\left[x, x^2\right]$$

```
%[n]
```

$$n^2$$

```
Map[Function[x, x^2], a + b + c]
```

$$a^2 + b^2 + c^2$$

```
Nest[Function[q, 1 / (1 + q)], x, 3]
```

$$\frac{1}{1+\frac{1}{1+\frac{1}{1+x}}}$$

```
Function[{x, y}, x^2 + y^3][a, b]
```

$$a^2 + b^3$$

```
Map[#^2 &, a + b + c]
```

$$a^2 + b^2 + c^2$$

```
f[##, ##] &[x, y]
```

```
f[x, y, x, y]
```

```
Apply[f[##2, #1] &, {{a, b, c}, {ap, bp}}, {1}]
```

```
{f[b, c, a], f[bp, ap]}
```

```
fromdigits[digits_] := Fold[(10 #1 + #2) &, 0, digits]
```

```
fromdigits[{1, 5, 2, 3, 4}]
```

```
15 234
```

Applying Functions Repeatedly

```
? Fold
```

Fold[*f*, *x*, *list*] gives the last element of FoldList[*f*, *x*, *list*]. ≫

```
? FoldList
```

FoldList[*f*, *x*, {*a*, *b*, …}] gives {*x*, *f*[*x*, *a*], *f*[*f*[*x*, *a*], *b*], …}. ≫

```
? Nest
```

Nest[*f*, *expr*, *n*] gives an expression with *f* applied *n* times to *expr*. ≫

```
? NestList
```

NestList[*f*, *expr*, *n*] gives a list of the results of applying *f* to *expr* 0 through *n* times. ≫

```
Nest[f, x, 4]
```

```
f[f[f[f[x]]]]
```

```
NestList[f, x, 4]
```

```
{x, f[x], f[f[x]], f[f[f[x]]], f[f[f[f[x]]]]}
```

```
recip[x_] := 1 / (1 + x)
```

```
Nest[recip, x, 3]
```

$$\frac{1}{1+\frac{1}{1+\frac{1}{1+x}}}$$

```
newton3[x_] := N[1 / 2 (x + 3 / x)]
```

```
NestList[newton3, 1.0, 5]
```

```
{1., 2., 1.75, 1.73214, 1.73205, 1.73205}
```

```
FixedPoint[newton3, 1.0]
```

```
1.73205
```

```
FixedPointList[newton3, 1.0]
```

```
{1., 2., 1.75, 1.73214, 1.73205, 1.73205, 1.73205}
```

```
divide2[n_] := n / 2
```

```
NestWhileList[divide2, 123 456, EvenQ]
```

```
{123 456, 61 728, 30 864, 15 432, 7716, 3858, 1929}
```

```
NestWhileList[newton3, 1.0, Unequal, 2]
```

```
{1., 2., 1.75, 1.73214, 1.73205, 1.73205, 1.73205}
```

```
NestWhileList[Mod[5 #, 7] &, 1, Unequal, All]
```

```
{1, 5, 4, 6, 2, 3, 1}
```

```
FoldList[f, x, {a, b, c}]
```

```
{x, f[x, a], f[f[x, a], b], f[f[f[x, a], b], c]}
```

```
Fold[f, x, {a, b, c}]
```

```
f[f[f[x, a], b], c]
```

```
FoldList[Plus, 0, {a, b, c}]
```

```
{0, a, a + b, a + b + c}
```

Testing and Searching List Elements

```
Position[{a, b, c, a, b}, a]
```

```
{{1}, {4}}
```

```
Count[{a, b, c, a, b}, a]
```

```
2
```

```
MemberQ[{a, b, c}, a]
```

```
True
```

```
MemberQ[{a, b, c}, d]
```

```
False
```

```
m = IdentityMatrix[3]
```

```
{{1, 0, 0}, {0, 1, 0}, {0, 0, 1}}
```

```
FreeQ[m, 0]
```

```
False
```

```
Position[m, 0]
```

```
{{1, 2}, {1, 3}, {2, 1}, {2, 3}, {3, 1}, {3, 2}}
```

Building Lists from Functions

```
Array[p, 5]
```

```
{p[1], p[2], p[3], p[4], p[5]}
```

```
Table[p[i], {i, 5}]
```

```
{p[1], p[2], p[3], p[4], p[5]}
```

```
Map[p, Range[5]]
```

```
{p[1], p[2], p[3], p[4], p[5]}
```

```
p[x_] := p^2
Array[p, 5]
Table[p[i], {i, 5}]
Map[p, Range[5]]
```

$$\{p^2, p^2, p^2, p^2, p^2\}$$

$$\{p^2, p^2, p^2, p^2, p^2\}$$

$$\{p^2, p^2, p^2, p^2, p^2\}$$

```
Array[# + #^2 &, 5]
```

```
{2, 6, 12, 20, 30}
```

```
Array[m, {2, 3}]
```

```
{{m[1, 1], m[1, 2], m[1, 3]}, {m[2, 1], m[2, 2], m[2, 3]}}
```

```
Array[Plus[##]^2 &, {3, 3}]
```

```
{{4, 9, 16}, {9, 16, 25}, {16, 25, 36}}
```

```
NestList[D[#, x] &, x^n, 3]
```

$$\left\{x^n,\ n\,x^{-1+n},\ (-1+n)\,n\,x^{-2+n},\ (-2+n)\,(-1+n)\,n\,x^{-3+n}\right\}$$

Selecting Parts of Expressions with Functions

```
Select[{2, 15, 1, a, 16, 17}, # > 4 &]
```

```
{15, 16, 17}
```

```
t = Expand[(x + y + z)^2]
```

$$x^2 + 2\,x\,y + y^2 + 2\,x\,z + 2\,y\,z + z^2$$

```
Select[t, FreeQ[#, x] &]
```

$y^2 + 2\,y\,z + z^2$

```
Select[{-1, 3, 10, 12, 14}, # > 3 &, 1]
```

{10}

Everything Is an Expression

```
x + y + z
```

x + y + z

```
FullForm[%]
```

```
Plus[x, y, z]
```

```
1 + x^2 + (y + z)^2
```

$1 + x^2 + (y + z)^2$

```
FullForm[%]
```

```
Plus[1, Power[x, 2], Power[Plus[y, z], 2]]
```

```
Head[f[x, y]]
```

f

```
? Head
```

Head[*expr*] gives the head of *expr*. ≫

```
Head[a + b + c]
```

```
Plus
```

```
Head[{a, b, c}]
```

```
List
```

```
Head[23 432]
```

```
Integer
```

```
Head[345.6]
```

```
Real
```

```
Part[{a, b, c}, 0]
```

```
List
```

```
Part[{a, b, c}, 1]
```

a

```
Part[{a, b, c}, 3]
```

c

```
Part[{a, b, c}, 4]
```

Part::partw : Part 4 of {a, b, c} does not exist. ≫

```
{a, b, c}[[4]]
```

```
Part[23 242, 0]
```

```
Integer
```

```
Part[23 242, 1]
```

Part::partd : Part specification 23242[[1]] is longer than depth of object. ≫

```
23 242[[1]]
```

Immediate and Delayed Definitions

```
ex[x_] := Expand[(1 + x)^2]
```

```
? ex
```

```
Global`ex

ex[x_] := Expand[(1 + x)^2]
```

```
iex[x_] = Expand[(1 + x)^2]
```

$1 + 2\,x + x^2$

```
? iex
```

```
Global`iex

iex[x_] = 1 + 2 x + x^2
```

```
ex[y + 2]
```

$9 + 6\,y + y^2$

```
iex[y + 2]
```

$1 + 2\,(2 + y) + (2 + y)^2$

```
r1 = RandomReal[]
```

```
0.0560708
```

```
r2 := RandomReal[]
```

```
{r1, r2}
```

```
{0.0560708, 0.6303}
```

```
{r1, r2}
```

```
{0.0560708, 0.359894}
```

Applying Transformation Rules

```
x + y /. x -> 3
```

$3 + y$

```
x + y /. {x -> a, y -> b}
```

```
a + b
```

```
x + y /. {{x -> 1, y -> 2}, {x -> 4, y -> 2}}
```

```
{3, 6}
```

```
Solve[x^3 - 5 x^2 + 2 x + 8 == 0, x]
```

```
{{x → -1}, {x → 2}, {x → 4}}
```

```
x^2 + 6 /. %
```

```
{7, 10, 22}
```

```
{x^2, x^3, x^4} /. {x^3 -> u, x^n_ -> p[n]}
```

```
{p[2], u, p[4]}
```

```
h[x + h[y]] /. h[u_] -> u^2
```

```
(x + h[y])^2
```

```
{x^2, y^3} /. {x -> y, y -> x}
```

```
{y^2, x^3}
```

```
x^2 /. x -> (1 + y) /. y -> b
```

```
(1 + b)^2
```

```
x^2 + y^6 /. {x -> 2 + a, a -> 3}
```

```
(2 + a)^2 + y^6
```

```
x^2 + y^6 //. {x -> 2 + a, a -> 3}
```

```
25 + y^6
```

```
log[a b c d] /. log[x_ y_] -> log[x] + log[y]
```

```
log[a] + log[b c d]
```

```
log[a b c d] //. log[x_ y_] -> log[x] + log[y]
```

```
log[a] + log[b] + log[c] + log[d]
```

THIS WAS A LOT OF MATERIAL - YOU MUST NOW PRACTICE.